

K. WRZEŚNIEWSKI and ELŻBIETA WINKEL (Poznań)

## DETERMINATION OF THE JORDAN CANONICAL FORM OF A REAL MATRIX

**1. Procedure declaration.** Procedure *jordanform1* finds the Jordan canonical form of a given real matrix  $A$  whose eigenvalues are known.

Data:

- $n$  — order of the matrix  $A$ ;
- $a[1:n, 1:n]$  — array containing the elements of the real matrix  $A$  (the matrix  $A$  is not destroyed in the process);
- $ne$  — number of distinct eigenvalues of the matrix  $A$ ;
- $lambda[1:ne]$  — array containing the distinct eigenvalues of  $A$ ;
- $eps$  — input parameter of procedure *matrixrank* (giving the criterion of the linear independence of the columns of the matrix  $A - \lambda I$ );
- $inprod$  — function with heading **real procedure** *inprod* ( $i, a, b, x, y$ ); **value**  $a, b$ ; **integer**  $i, a, b$ ; **real**  $x, y$ ; which for  $i = a, a+1, \dots, b$  accumulates the sum of products  $x \times y$  in double precision;
- $matrixrank$  — function with heading **integer procedure** *matrixrank* ( $n, a, eps$ ); **value**  $n$ ; **integer**  $n$ ; **real**  $eps$ ; **array**  $a$ ; which calculates the rank of a real matrix.

Results:

- $e[1:n]$  — array of type **integer** the elements from  $e[2]$  to  $e[n]$  of which give  $n-1$  superdiagonal elements of the Jordan canonical form of the matrix  $A$ ; the element  $e[1]$  is set to be zero;
- $d[1:n]$  — array giving the diagonal elements of the Jordan canonical form of the matrix  $A$ .

**2. Method used.** In general, the structure of the Jordan canonical form of a matrix  $A$  can be described as follows.

```

procedure jordanform1(n,a,ne,lambda,eps,inprod,matrixrank)
  result:(e,d);
  value n;
  integer n,ne;
  real eps;
  integer array e;
  array a,lambda,d;
  real procedure inprod;
  integer procedure matrixrank;
  begin
    real lami;
    integer i,j,k,l,m,r,r1,r2,q1,q2,y;
    integer array s[1:n];
    array b,c[1:n,1:n],p[1:n];
    for i:=1 step 1 until n do
      begin
        p[i]:=a[i,i];
        e[i]:=0
      end i;
    y:=0;
    for i:=1 step 1 until ne do
      begin
        lami:=lambda[i];
        for j:=1 step 1 until n do
          for k:=1 step 1 until n do
            if j=k
              then a[j,k]:=b[j,k]:=c[j,k]:=p[j]-lami
              else b[j,k]:=c[j,k]:=a[j,k];
          r1:=matrixrank(n,c,eps);
          q1:=q2:=n-r1;

```

```

j:=0;
for j:=j+1 while q2>0 do
  begin
    for k:=1 step 1 until n do
      for l:=1 step 1 until n do
        c[l,k]:=inprod(m,1,n,a[l,m],b[m,k]);
      for k:=1 step 1 until n do
        for l:=1 step 1 until n do
          b[k,l]:=c[k,l];
        r2:=matrixrank(n,c,eps);
        q2:=r1-r2;
        r1:=r2;
        r:=s[j]:=q1-q2;
        q1:=q2;
        for k:=1 step 1 until r do
          begin
            m:=y+j-1;
            for l:=y+1 step 1 until m do
              begin
                e[l+1]:=1;
                d[l]:=lami
              end l;
            y:=y+j;
            d[y]:=lami
          end k
        end j
      end i;
    for i:=1 step 1 until n do
      a[i,i]:=p[i]
    end jordanform1
  end

```

The Jordan canonical form of a matrix  $A$  is a quasidiagonal matrix

$$J = \begin{bmatrix} J_1 & & & 0 \\ & \ddots & & \\ & & J_i & \\ 0 & & & \ddots \\ & & & & J_m \end{bmatrix}$$

composed of canonical boxes  $J_i$ , corresponding to distinct eigenvalues of  $A$ , of the form

$$J_i = \begin{bmatrix} J_{i1} & & & 0 \\ & \ddots & & \\ & & J_{ij} & \\ 0 & & & \ddots \\ & & & & J_{im_i} \end{bmatrix},$$

where

$$J_{ij} = \begin{bmatrix} \lambda_i & & 1 & & 0 \\ & \ddots & & \ddots & \\ & & \ddots & & \\ 0 & & & \ddots & \\ & & & & \lambda_i & \\ & & & & & 1 \end{bmatrix}$$

is an elementary Jordan block of order  $k_{ij}$ ,

$$\sum_{j=1}^{m_i} k_{ij} = k_i,$$

and  $\lambda_i$  is an eigenvalue of  $A$  of multiplicity  $k_i$ . The number of distinct eigenvalues is  $m$ , and

$$\sum_{i=1}^m k_i = n$$

is the order of  $A$ .

In the Jordan canonical form, built by means of procedure *jordanform1*, elementary Jordan blocks  $J_{ij}$  are arranged according to non-decreasing orders of  $k_{ij}$ .

The method used in procedure *jordanform1* is based on the following theorem stated and proved in [4]:

**THEOREM.** *Let  $A$  be a matrix of complex numbers of order  $n$ , and suppose that  $\lambda$  is an eigenvalue of  $A$ . For each  $j = 1, 2, \dots, p$  let  $s_j$  be the number*

of elementary Jordan blocks of order  $j$  in the canonical box associated with  $\lambda$  of the Jordan canonical form of  $A$ . Then  $s_j$  is given by

$$s_j = r_{j+1} - 2r_j + r_{j-1}, \quad j = 1, 2, \dots, p,$$

where

$$r_j = \text{rank } B^j, \quad j = 0, 1, \dots, p+1, \quad B = A - \lambda I.$$

The dimension of the largest elementary Jordan block is the smallest non-negative integer  $p$  for which  $r_p = r_{p+1}$ .

By convention, let  $B^0 = I$  for any square matrix  $B$ . Hence  $r_0 = \text{rank } B^0 = \text{rank } I = n$ .

On the basis of the Theorem the following algorithm has been built to determine the structure of the Jordan canonical form of the matrix  $A$ .

For each distinct eigenvalue  $\lambda_i$  repeat the following steps:

Step 1. Compute  $B = A - \lambda_i I$ .

Step 2. Compute  $r_1 = \text{rank } B$  and the first decrease of rank  $q_1 = n - r_1$ .

Step 3. For each  $j = 1, 2, \dots$  repeat the following three steps terminating as soon as the condition  $r_j = r_{j+1}$  (i.e.  $q_{j+1} = 0$ ) is satisfied.

Step 3.1. Compute  $r_{j+1} = \text{rank } B^{j+1}$  and the decrease of rank  $q_{j+1} = r_j - r_{j+1}$ .

Step 3.2. Compute the  $j$ -th element of the characteristic  $s_j = q_j - q_{j+1}$ .

Step 3.3. Build  $s_j$  elementary Jordan blocks of order  $j$ .

**3. Applicability and certification.** Procedure *jordanform1* is designed to find in a simple way the Jordan canonical form of a real matrix  $A$  with known real eigenvalues. The correctness of the calculation of the Jordan canonical form depends on the accuracy with which the eigenvalues of  $A$  are computed. The eigenvalues of the testing matrices were calculated by means of the set of procedures *balance*, *dirhes*, and *hqr* inserted in [5].

In the course of testing calculations, the library procedure *inprod* described in [3] has been used. Furthermore, procedure *matrixrank* containing procedure *svd*, inserted in [1] and [5], has been used. In the testing program, as the input parameter *eps* the estimation  $\text{eps} = 2^{-t/2+3} \|A\|_E$  has been used, where  $t$  denotes the number of binary digits of the mantissa during the calculation of eigenvalues.

Before using procedure *jordanform1*, matrices were scaled to satisfy the condition  $0.1n \leq \|A\|_E \leq 10n$ .

Procedure *jordanform1* has extensively been tested on the Odra 1305 computer. The results obtained were correct.

As an illustration, the result obtained for an example taken from [2], p. 110-113, is given.

Data:

$$n = 10,$$

$$a = \begin{bmatrix} 1 & 1 & 1 & -2 & 1 & -1 & 2 & -2 & 4 & -3 \\ -1 & 2 & 3 & -4 & 2 & -2 & 4 & -4 & 8 & -6 \\ -1 & 0 & 5 & -5 & 3 & -3 & 6 & -6 & 12 & -9 \\ -1 & 0 & 3 & -4 & 4 & -4 & 8 & -8 & 16 & -12 \\ -1 & 0 & 3 & -6 & 5 & -4 & 10 & -10 & 20 & -15 \\ -1 & 0 & 3 & -6 & 2 & -2 & 12 & -12 & 24 & -18 \\ -1 & 0 & 3 & -6 & 2 & -5 & 15 & -13 & 28 & -21 \\ -1 & 0 & 3 & -6 & 2 & -5 & 12 & -11 & 32 & -24 \\ -1 & 0 & 3 & -6 & 2 & -5 & 12 & -14 & 37 & -26 \\ -1 & 0 & 3 & -6 & 2 & -5 & 12 & -14 & 36 & -25 \end{bmatrix},$$

$$ne = 3, \quad \lambda = [1.0 \ 2.0 \ 3.0], \quad \epsilon = 5.05_{10}^{-8}.$$

Results:

$$e = [0 \ 0 \ 1 \ 0 \ 1 \ 1 \ 0 \ 1 \ 0 \ 1],$$

$$d = [1 \ 2 \ 2 \ 2 \ 2 \ 2 \ 3 \ 3 \ 3 \ 3].$$

Time of computation was equal to 11 seconds.

#### References

- [1] G. H. Golub and C. Reinsch, *Singular value-decomposition and least squares solutions*, Numer. Math. 14 (1970), p. 403-420.
- [2] R. T. Gregory and D. L. Karney, *A collection of matrices for testing computational algorithms*, J. Wiley, New York 1969.
- [3] *Procedura inprod: obliczanie z podwójną precyzją sumy ciągu liczb, z których każda jest iloczynem dwóch liczb*, Oprogramowanie maszyn cyfrowych ODRA serii 1300, ALGOL, Procedury pomocnicze, Zeszyt 1, p. 73, Elwro, Wrocław 1973.
- [4] W. B. Rubin, *A simple method for finding the Jordan form of a matrix*, IEEE Trans. Autom. Control AC-17 (1972), p. 145-146.
- [5] J. H. Wilkinson and C. Reinsch, *Linear algebra*, in: *Handbook for automatic computation*, p. 315-326, 339-371, vol. 2, Springer, Berlin 1971.

TECHNICAL UNIVERSITY  
60-965 POZNAŃ

Received on 1. 7. 1974;  
revised version on 17. 4. 1976

K. WRZEŚNIEWSKI i ELŻBIETA WINKEL (Poznań)

ALGORYTM 60

## OBLICZENIE POSTACI KANONICZNEJ JORDANA MACIERZY RZECZYWISTEJ

## STRESZCZENIE

Procedura *jordanform1* znajduje postać kanoniczną Jordana danej macierzy rzeczywistej o znanym rzeczywistym spektrum.

Dane:

- $n$  — stopień macierzy  $A$ ;
- $a[1:n, 1:n]$  — tablica zawierająca elementy macierzy rzeczywistej  $A$  (macierz  $A$  nie zostaje zniszczona w trakcie obliczeń);
- $ne$  — liczba różnych wartości własnych macierzy  $A$ ;
- $lambda[1:ne]$  — tablica zawierająca różne wartości własne macierzy  $A$ ;
- $eps$  — parametr wejścia procedury *matrixrank* (podaje kryterium liniowej niezależności kolumn macierzy  $A - \lambda I$ );
- inprod* — funkcja o nagłówku **real procedure inprod**( $i, a, b, x, y$ ); **value**  $a, b$ ; **integer**  $i, a, b$ ; **real**  $x, y$ ; która dla  $i = a, a + 1, \dots, b$  kumuluje sumę iloczynów  $x \times y$  w podwójnej precyzji;
- matrixrank* — funkcja o nagłówku **integer procedure matrixrank**( $n, a, eps$ ); **value**  $n$ ; **integer**  $n$ ; **real**  $eps$ ; **array**  $a$ ; która oblicza rząd macierzy rzeczywistej.

Wyniki:

- $e[1:n]$  — tablica typu **integer**, której elementy od  $e[2]$  do  $e[n]$  zawierają  $n - 1$  ponaddiagonalnych elementów postaci kanonicznej Jordana macierzy  $A$ ; element  $e[1] = 0$ ;
- $d[1:n]$  — tablica zawierająca elementy diagonalne postaci kanonicznej Jordana macierzy  $A$ .

W procedurze *jordanform1* zastosowano metodę opisaną w § 2, opartą na twierdzeniu sformułowanym i udowodnionym w [4].

W trakcie wykonywania obliczeń korzystano z procedury bibliotecznej *inprod*, opisaney w [3], oraz z procedury *matrixrank*, zawierającej procedurę *svd* zamieszczonej w [1] i [5].

Procedura *jordanform1* była sprawdzana na wielu przykładach na maszynie cyfrowej Odra 1305. Otrzymane wyniki były poprawne.